

Gramática de nuestro lenguaje que cumple las reglas LL(1)

Los símbolos en **azul** son los tokens o lexemas que se deben almacenar en la tabla de símbolos (puede ser una lista de tokens).

```
<PROGRAMA>::= comienza <LISTASENTENCIAS> termina
<LISTASENTENCIAS>::= <SENTENCIA> <LSP>
<LSP>::= <SENTENCIA> <LSP> | ∅
<SENTENCIA>::= <DECLVAR> | <CICLOMIENTRAS> | <SI> | <CONID>
| <ASIGNACION> | <IMPRIMIR> | <LEER>
<DECLVAR>::= <TIPO> <LISTAID>;
<TIPO>::= entero | flotante
<LISTAID>::= <ID> <LIP>
<LIP>::= , <ID> <LIP> | ∅
<ID>::= ([a-z] | [A-Z]) ([a-z] | [A-Z] | [0-9])*
<EXPARIT>::= <TERMINO> <EAP>
<EAP>::= + <TERMINO> <EAP> | - <TERMINO> <EAP> | ∅
<TERMINO>::= <FACTOR> <TP>
<TP>::= * <FACTOR> <TP> | / <FACTOR> <TP> | ∅
<FACTOR>::= ( <EXP_ARIT> ) | <ID> | <NUMERO>
<NUMERO>::= <NENTERO> | <NFLOTANTE>
<NENTERO>::= 0 | [1-9][0-9]*
<NFLOTANTE>::= <NENTERO>.[0-9]+
<CICLOMIENTRAS>::= mientras ( <CONDICION> ) { <LISTASENTENCIAS> }
<CONDICION>::= <EXP_ARIT> <CP>
<CP>::= < <EXPARIT> | <= <EXPARIT> | > <EXPARIT> | >= <EXPARIT> | ==
<EXPARIT>
| != <EXPARIT>
<SI>::= si ( <CONDICION> ) { <LISTASENTENCIAS> }
<CONID>::= con ( <ID> ) { <LISTACASOS> pordefecto : <LISTASENTENCIAS> }
<LISTACASOS>::= caso <NUMERO> : <LISTASENTENCIAS> rompe ;
<LISTACASOS> | ∅
<ASIGNACION>::= <ID> = <EXPARIT> ;
<IMPRIMIR>::= imprime ( <CADENA> ) ;
<CADENA>::= <TEXTO> <CAP> | <EXPARIT> <CAP>
<CAP>::= & <CAPP> | ∅
<CAPP>::= <EXPARIT> <CAPP> | <TEXTO> <CAPP> | ∅
<TEXTO>::= "([a-z] | [A-Z] | [0-9] | + | - | * | / | . | : | , | ; | ( | ) )*"
<LEER>::= lee ( <ID> ) ;
```

Lexemas a identificar

Palabras reservadas

- ❖ comienza
- ❖ termina
- ❖ entero
- ❖ flotante
- ❖ mientras
- ❖ si
- ❖ con
- ❖ pordefecto
- ❖ caso
- ❖ rompe
- ❖ imprime
- ❖ lee

Símbolos

- ❖ Suma +
- ❖ Resta -
- ❖ Multiplicación *
- ❖ División /
- ❖ Coma ,
- ❖ ParAbre (
- ❖ ParCierra)
- ❖ LlaveAbre {
- ❖ LlaveCierra }
- ❖ DosPuntos :
- ❖ PuntoYComa ;
- ❖ Menor <
- ❖ MenorIgual <=
- ❖ Mayor >
- ❖ MayorIgual >=
- ❖ Igual ==
- ❖ Diferente !=
- ❖ Asignación =
- ❖ Concatena &

Otros

- ❖ ID
- ❖ TEXTO
- ❖ NENTERO
- ❖ NFLOTANTE